

## Program Corner

### การสร้างกราฟิก FBD คานต่อเนื่อง ตอนที่ 1

ผศ.สรกานต์ ศรีทองอ่อน

เมื่อเรากำหนดสเกลของโปรแกรมให้เป็นพิกัดโลก (ในระบบ 2 มิติ) แล้ว ต่อไปการสร้างกราฟิกในระบบ 2 มิติ ก็ไม่ใช่เรื่องยาก ซึ่งในตอนนี้จะแสดงถึง การสร้างกราฟิกแผนภาพวัตถุอิสระ (free body diagram : FBD) ของคานต่อเนื่อง (continuous beam) ในตอนที่ 1 นะครับ

#### แนวคิด

- การแสดง FBD ของคานต่อเนื่อง ใช้เป็นรูปสี่เหลี่ยมต่อเนื่องกันไปในแต่ละช่วงคาน โดยใช้คำสั่ง

**Line(X1, Y1) - (X2, Y2), QBColor(รหัสสี), B**

โดย

X1, Y1 คือ พิกัดมุมบนซ้ายของคานแต่ละช่วง

X2, Y2 คือ พิกัดมุมล่างขวาของคานแต่ละช่วง

- การลากเส้นบอกความยาวของคานในแนวนอน ใช้ลากเส้นจากปลายคานช่วงซ้ายสุด ไปจนถึงปลายคานช่วงขวาสุด โดยใช้คำสั่ง

**Line(X, Y) - (X+SumL, Y), QBColor(รหัสสี)**

โดย

X คือ พิกัดแกน X ของปลายคานช่วงซ้ายสุด โดยในที่นี้ให้เท่ากับ 0

Y คือ พิกัดแกน Y ของเส้นบอกความยาวคานในแนวนอน โดยให้ต่ำลงมาจากรูปคาน ในที่นี้กำหนดให้ต่ำลงมา 2 หน่วย

SumL คือ ผลรวมของความยาวคานแต่ละช่วง

- จิตเส้นบอกระยะแนวตั้ง โดยจำนวนเส้นเท่ากับ จำนวนช่วงคาน (N)+1 ใช้คำสั่ง

**Line(X, Y+0.5) - (X, Y-0.5)**

โดย

- X คือ พิกัดแกน X ด้านปลายของคานแต่ละช่วง (เริ่มต้นลากเส้นที่ปลายซ้ายของคานช่วงแรก ต่อมาลากเส้นที่ปลายขวาของคาน)
- Y คือ พิกัดแกน Y ของเส้นบอกความยาวคานในแนวนอน
- 0.5 คือ ระยะจาก Y ขึ้นและลงแนวละ 0.5 หน่วย

- ตำแหน่งในการบอกระยะความยาวของแต่ละช่วงคาน จะเซตให้อยู่กึ่งกลางของแต่ละช่วงคานนั้น (ดูรายละเอียดจากโปรแกรม)

### โปรแกรม

- จากโปรแกรมในบทความตอนที่แล้ว (การกำหนดสเกลในการแสดงผลที่จอภาพ) สร้าง Sub เพิ่ม ให้ชื่อว่า PlotBeam แล้วพิมพ์คำสั่งดังภาพที่ 1 และภาพที่ 2

```

Sub PlotBeam()
Const N = 3 ' สมมุติจำนวนช่วงคานเท่ากับ 3
Dim L(1 To N) As Single ' ความยาวแต่ละช่วงคาน
Dim SumL As Single ' ผลรวมความยาวของแต่ละช่วงคาน
Dim I As Integer
Dim X As Single: Dim Y As Single
Dim d As String

L(1) = 6: L(2) = 3.5: L(3) = 5.75 ' สมมุติความยาวแต่ละช่วงคาน
' หาผลรวมความยาวของแต่ละช่วงคาน
SumL = 0
For I = 1 To N
    SumL = SumL + L(I)
Next I
' สร้างกราฟคานแต่ละช่วง
X = 0: Y = 10
For I = 1 To N
    Line (X, Y)-(X + L(I), Y + 0.1), QBColor(1), B
    X = X + L(I)
Next I

```

ภาพที่ 1 คำสั่งใน Sub PlotBeam

```


' เส้นบอกความยาวแต่ละช่วง
X = 0: Y = 8
Line (X, Y)-(X + SumL, Y), QBColor(4)
Line (X, Y + 0.5)-(X, Y - 0.5), QBColor(4)
For I = 1 To 3
    X = X + L(I)
    Line (X, Y + 0.5)-(X, Y - 0.5), QBColor(4)
Next I
' แสดงค่าความยาวแต่ละช่วง
X = 0: Y = 8
For I = 1 To 3
    d = Format(L(I), '0.00')
    CurrentX = X + (L(I) / 2) - (TextWidth(d) / 2)
    CurrentY = Y - TextHeight(d)
    Print d
    X = X + L(I)
Next I
End Sub

```

ภาพที่ 2 คำสั่งใน Sub PlotBeam (ต่อ)


ใน Sub นี้ สมมุติว่าพล็อต 3 ช่วงแกน มีความยาวตามที่สมมุติในโปรแกรม ซึ่งผู้อ่านลองเปลี่ยนค่า แล้วสังเกตการแสดงผลดูนะครับ ส่วนคำสั่งอื่นๆ เป็นคำสั่งทั่วไปนำไปใช้ได้เลย

- การเรียกใช้ Sub PlotBeam เรียกที่เหตุการณ์ (event) Form\_Activate โดยพิมพ์คำสั่งต่อท้าย คำสั่งเดิมจากบทความตอนที่แล้ว (การกำหนดสเกลในการแสดงผลที่จอภาพ) ดังภาพที่ 3



ภาพที่ 3 พิมพ์คำสั่ง PlotBeam ต่อท้าย ในเหตุการณ์ Form\_Activate

- เมื่อรันโปรแกรม จะแสดงผลดังภาพที่ 4



ภาพที่ 4 กราฟิก FBD คาน

- ❑ คำสั่งตอนนี้ยังไม่ได้พิจารณากรณีความยาวรวมของคานเกินกรอบจอภาพ
- ❑ กรณีแสดงที่รองรับ (support) ประกอบในกราฟิก FBD คาน จะมีความซับซ้อนขึ้นในแง่ว่าเป็นคานยื่นหรือไม่ ซึ่งจะได้นำเสนอในตอนต่อไป

### หนังสืออ้างอิง

สรกานต์ ศรีทองอ่อน. วิชาการเบสิกในงานวิศวกรรมโยธา. กรุงเทพฯ : สำนักพัฒนาเทคนิคศึกษา สถาบันเทคโนโลยีพระจอมเกล้าพระนครเหนือ, 2544.

สรกานต์ ศรีทองอ่อน. “การกำหนดสเกลในการแสดงผลที่จอภาพ”. วารสารพัฒนาเทคนิคศึกษา ปีที่ \_\_ ฉบับที่ \_\_\_\_\_ , บริษัท 21 เซ็นจูรี จำกัด, 25 \_\_.

Weiskamp, Keith and Heiny, Loren. Power Graphics Using Turbo Pascal 6. the United States of America : John Wiley & Sons, Inc., 1991.

เว็บไซต์ [www.geocities.com/karnprogram](http://www.geocities.com/karnprogram).